

1. 地域貿易協定の影響

重力方程式(gravity equation)を用いて、地域貿易協定が貿易に与える影響を研究することができる。ここでは、北米自由貿易協定 (NAFTA) によって、加盟国 (アメリカ、カナダ、メキシコ) の貿易がどのように変化したのかを調べる。データは、agGravityDatasmall.dta を用いる。

```
use "agGravityDatasmall.dta", clear
```

このデータには、1976 年から 2004 年までの 2 国間の貿易データが収録されている。

2. 加盟国の識別

まず加盟国であるカナダ、メキシコ、アメリカを識別するダミー変数を作成する。

* NAFTA 輸入国ダミー

```
gen nafta = (ccode=="CAN" | ccode=="MEX" | ccode=="USA")  
label var nafta "1 if home is nafta member"
```

* NAFTA 輸出国ダミー

```
gen pnafta = (pcode=="CAN" | pcode=="MEX" | pcode=="USA")  
label var pnafta "1 if partner is nafta member"
```

3. 加盟国の貿易

NAFTA 加盟国の貿易を (1) NAFTA 内の貿易、(2) NAFTA 加盟国が加盟国以外の国から輸入する貿易、(3) NAFTA 加盟国が加盟国以外の国へ輸出する貿易の三つそれぞれに識別するダミー変数を作成する。

* 1994 年の NAFTA 発効以降の NAFTA 加盟国内貿易を 1 とするダミー変数

```
gen intra_nafta = (ccode=="CAN" | ccode=="MEX" | ccode=="USA") & (pcode=="CAN"  
| pcode=="MEX" | pcode=="USA")  
replace intra_nafta = 0 if year < 1994  
label var intra_nafta "1 if trade between nafta members"
```

* NAFTA 加盟国が加盟国以外の国から輸入する場合を 1 とするダミー変数

```
gen imp_nafta_rest = (ccode=="CAN" | ccode=="MEX" | ccode=="USA") &  
(pcode!="CAN" & pcode!="MEX" & pcode!="USA")
```

```

replace imp_nafta_rest = 0 if year < 1994
label var imp_nafta_rest "1 if nafta's imports from the rest of the world"
* NAFTA 加盟国が加盟国以外の国へ輸出する場合を 1 とするダミー変数
gen exp_nafta_rest = (pcode=="CAN" | pcode=="MEX" | pcode=="USA") &
(ccode!="CAN" & ccode!="MEX" & ccode!="USA")
replace exp_nafta_rest = 0 if year < 1994
label var exp_nafta_rest "1 if nafta's exports to the rest of the world"

```

4. データの準備

重力方程式の推定に備えて、国名コードの数値化を行っておく。

```

encode pcode,gen(pcoden)
encode ccode,gen(ccoden)

```

また、パネルの識別子を作成する。ここでは、国のペアがパネルの識別子である。

```

egen id = group(ccode pcode)
tsset id year

```

さらに、変数の対数化を行う。

```

gen lnV = log(imp_tv)
label var lnV "value of imported goods in logarithm"
gen lncGDP = log(cgdp_current)
label var lncGDP "partner's current GDP in logarithm"
gen lnpGDP = log(pgdp_current)
label var lnpGDP "home's current GDP in logarithm"
gen lnD = log(km)
label var lnD "bilateral distance in logarithm"

```

5. 推定

推定に用いる 2 国間の貿易データは、1976 年から 2004 年までのパネルデータである。そのため、貿易国のペアの固定効果を考慮した固定効果モデルの推定を行う。

```

xtreg lnV lncGDP lnpGDP intra_nafta imp_nafta_rest exp_nafta_rest i.year, fe vce(robust)

```

ここで、従属変数は2国間の貿易額 (lnV) である。また説明変数には、輸入国の GDP の対数値 (lncGDP)、輸出国の GDP の対数値 (lnpGDP) を含めている。NAFTA 加盟の効果を測るため、NAFTA 域内貿易ダミー (intra_nafta)、NAFTA 加盟国の域外輸入ダミー (imp_nafta_rest)、NAFTA 加盟国の域外輸出ダミー (exp_nafta_rest) を加えている。貿易国のペアの固定効果を考慮した固定効果モデルのため、貿易国間の距離や言語の共通性といった変数は加えても固定効果に吸収されるため、加えていない。

```
Fixed-effects (within) regression      Number of obs   =   211,406
Group variable: id                   Number of groups =   15,728

R-sq:                                Obs per group:
    within = 0.1872                    min =          1
    between = 0.5011                   avg =         13.4
    overall = 0.4549                   max =          29

corr(u_i, Xb) = 0.2225                 F(33,15727)     =   306.78
                                         Prob > F        =   0.0000
```

(Std. Err. adjusted for 15,728 clusters in id)

lnV	Coef.	Robust Std. Err.	t	P> t	[95% Conf. Interval]	
lncGDP	.8587733	.0307071	27.97	0.000	.7985838	.9189627
lnpGDP	.55351	.0282126	19.62	0.000	.49821	.6088101
intra_nafta	.5023271	.2440053	2.06	0.040	.0240488	.9806055
imp_nafta_rest	.2395259	.0674902	3.55	0.000	.1072373	.3718145
exp_nafta_rest	-.02912	.0508948	-0.57	0.567	-.1288796	.0706395
year						
1977	.0510893	.0424041	1.20	0.228	-.0320275	.1342062
1978	-.0740576	.0458673	-1.61	0.106	-.1639628	.0158476
1979	-.092856	.0490368	-1.89	0.058	-.1889738	.0032619

(推定結果抜粋)

NAFTA 域内貿易ダミーが正に有意であることから、NAFTA によって、域内の貿易が増加したと考えられる。これは「貿易創出効果」に該当する。

一方で、NAFTA 加盟国の域外輸入ダミーも正に有意であることから、NAFTA の域外からの輸入も増えていることがわかる。さらに、NAFTA 加盟国の域外輸出ダミーは統計的に有意ではない。これらの結果からは、NAFTA の発効が NAFTA 加盟国と域外との貿易を阻害する「貿易転換効果」はなかったと考えられる。